

NAG Library Routine Document

F04JGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F04JGF finds the solution of a linear least squares problem, $Ax = b$, where A is a real m by n ($m \geq n$) matrix and b is an m element vector. If the matrix of observations is not of full rank, then the minimal least squares solution is returned.

2 Specification

```

SUBROUTINE F04JGF (M, N, A, LDA, B, TOL, SVD, SIGMA, IRANK, WORK, LWORK,      &
                  IFAIL)
INTEGER           M, N, LDA, IRANK, LWORK, IFAIL
REAL (KIND=nag_wp) A(LDA,N), B(M), TOL, SIGMA, WORK(LWORK)
LOGICAL          SVD

```

3 Description

The minimal least squares solution of the problem $Ax = b$ is the vector x of minimum (Euclidean) length which minimizes the length of the residual vector $r = b - Ax$.

The real m by n ($m \geq n$) matrix A is factorized as

$$A = Q \begin{pmatrix} R \\ 0 \end{pmatrix}$$

where Q is an m by m orthogonal matrix and R is an n by n upper triangular matrix. If R is of full rank, then the least squares solution is given by

$$x = (R^{-1} \ 0)Q^T b.$$

If R is not of full rank, then the singular value decomposition of R is obtained so that R is factorized as

$$R = UDV^T,$$

where U and V are n by n orthogonal matrices and D is the n by n diagonal matrix

$$D = \text{diag}(\sigma_1, \sigma_2, \dots, \sigma_n),$$

with $\sigma_1 \geq \sigma_2 \geq \dots \geq \sigma_n \geq 0$, these being the singular values of A . If the singular values $\sigma_{k+1}, \dots, \sigma_n$ are negligible, but σ_k is not negligible, relative to the data errors in A , then the rank of A is taken to be k and the minimal least squares solution is given by

$$x = V \begin{pmatrix} S^{-1} & 0 \\ 0 & 0 \end{pmatrix} \begin{pmatrix} U^T & 0 \\ 0 & I \end{pmatrix} Q^T b,$$

where $S = \text{diag}(\sigma_1, \sigma_2, \dots, \sigma_k)$.

The routine also returns the value of the standard error

$$\begin{aligned} \sigma &= \sqrt{\frac{r^T r}{m-k}}, & \text{if } m > k, \\ &= 0, & \text{if } m = k, \end{aligned}$$

$r^T r$ being the residual sum of squares and k the rank of A .

4 References

Lawson C L and Hanson R J (1974) *Solving Least Squares Problems* Prentice–Hall

5 Arguments

- 1: M – INTEGER *Input*
On entry: m , the number of rows of A .
Constraint: $M \geq N$.
- 2: N – INTEGER *Input*
On entry: n , the number of columns of A .
Constraint: $1 \leq N \leq M$.
- 3: A(LDA,N) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the m by n matrix A .
On exit: if SVD is returned as .FALSE., A is overwritten by details of the QR factorization of A .
 If SVD is returned as .TRUE., the first n rows of A are overwritten by the right-hand singular vectors, stored by rows; and the remaining rows of the array are used as workspace.
- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F04JGF is called.
Constraint: $LDA \geq M$.
- 5: B(M) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the right-hand side vector b .
On exit: the first n elements of B contain the minimal least squares solution vector x . The remaining $m - n$ elements are used for workspace.
- 6: TOL – REAL (KIND=nag_wp) *Input*
On entry: a relative tolerance to be used to determine the rank of A . TOL should be chosen as approximately the largest relative error in the elements of A . For example, if the elements of A are correct to about 4 significant figures then TOL should be set to about 5×10^{-4} . See Section 9 for a description of how TOL is used to determine rank. If TOL is outside the range $(\epsilon, 1.0)$, where ϵ is the **machine precision**, then the value ϵ is used in place of TOL. For most problems this is unreasonably small.
- 7: SVD – LOGICAL *Output*
On exit: is returned as .FALSE. if the least squares solution has been obtained from the QR factorization of A . In this case A is of full rank. SVD is returned as .TRUE. if the least squares solution has been obtained from the singular value decomposition of A .
- 8: SIGMA – REAL (KIND=nag_wp) *Output*
On exit: the standard error, i.e., the value $\sqrt{r^T r / (m - k)}$ when $m > k$, and the value zero when $m = k$. Here r is the residual vector $b - Ax$ and k is the rank of A .

- 9: IRANK – INTEGER *Output*
On exit: k , the rank of the matrix A . It should be noted that it is possible for IRANK to be returned as n and SVD to be returned as .TRUE.. This means that the matrix R only just failed the test for nonsingularity.
- 10: WORK(LWORK) – REAL (KIND=nag_wp) array *Output*
On exit: if SVD is returned as .FALSE., then the first n elements of WORK contain information on the QR factorization of A (see argument A above), and $WORK(n + 1)$ contains the condition number $\|R\|_E \|R^{-1}\|_E$ of the upper triangular matrix R .
 If SVD is returned as .TRUE., then the first n elements of WORK contain the singular values of A arranged in descending order and $WORK(n + 1)$ contains the total number of iterations taken by the QR algorithm. The rest of WORK is used as workspace.
- 11: LWORK – INTEGER *Input*
On entry: the dimension of the array WORK as declared in the (sub)program from which F04JGF is called.
Constraint: $LWORK \geq 4 \times N$.
- 12: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, LDA = $\langle value \rangle$ and M = $\langle value \rangle$.

Constraint: LDA \geq M.

On entry, LWORK is too small. Minimum size required: $\langle value \rangle$.

On entry, M = $\langle value \rangle$ and N = $\langle value \rangle$.

Constraint: M \geq N.

On entry, N = $\langle value \rangle$.

Constraint: N \geq 1.

IFAIL = 2

The QR algorithm has failed to converge to the singular values in $50 \times N$ iterations. This failure can only happen when the singular value decomposition is employed, but even then it is not likely to occur.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed factors Q , R , U , D and V^T satisfy the relations

$$Q \begin{pmatrix} R \\ 0 \end{pmatrix} = A + E,$$

$$Q \begin{pmatrix} U & 0 \\ 0 & I \end{pmatrix} \begin{pmatrix} D \\ 0 \end{pmatrix} V^T = A + F,$$

where

$$\|E\|_2 \leq c_1 \epsilon \|A\|_2,$$

$$\|F\|_2 \leq c_2 \epsilon \|A\|_2,$$

ϵ being the *machine precision*, and c_1 and c_2 being modest functions of m and n . Note that $\|A\|_2 = \sigma_1$.

For a fuller discussion, covering the accuracy of the solution x see Lawson and Hanson (1974), especially pages 50 and 95.

8 Parallelism and Performance

F04JGF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F04JGF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

If the least squares solution is obtained from the QR factorization then the time taken by the routine is approximately proportional to $n^2(3m - n)$. If the least squares solution is obtained from the singular value decomposition then the time taken is approximately proportional to $n^2(3m + 19n)$. The approximate proportionality factor is the same in each case.

This routine is column biased and so is suitable for use in paged environments.

Following the QR factorization of A the condition number

$$c(R) = \|R\|_E \|R^{-1}\|_E$$

is determined and if $c(R)$ is such that

$$c(R) \times \text{TOL} > 1.0$$

then R is regarded as singular and the singular values of A are computed. If this test is not satisfied, R is regarded as nonsingular and the rank of A is set to n . When the singular values are computed the rank of A , say k , is returned as the largest integer such that

$$\sigma_k > \text{TOL} \times \sigma_1,$$

unless $\sigma_1 = 0$ in which case k is returned as zero. That is, singular values which satisfy $\sigma_i \leq \text{TOL} \times \sigma_1$ are regarded as negligible because relative perturbations of order TOL can make such singular values zero.

10 Example

This example obtains a least squares solution for $r = b - Ax$, where

$$A = \begin{pmatrix} 0.05 & 0.05 & 0.25 & -0.25 \\ 0.25 & 0.25 & 0.05 & -0.05 \\ 0.35 & 0.35 & 1.75 & -1.75 \\ 1.75 & 1.75 & 0.35 & -0.35 \\ 0.30 & -0.30 & 0.30 & 0.30 \\ 0.40 & -0.40 & 0.40 & 0.40 \end{pmatrix}, \quad b = \begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \\ 5 \\ 6 \end{pmatrix}$$

and the value TOL is to be taken as 5×10^{-4} .

10.1 Program Text

```

Program f04jgfe

!      F04JGF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: f04jgfe, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: sigma, tol
Integer                    :: i, ifail, irank, lda, lwork, m, n
Logical                    :: svd
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:, :), b(:), work(:)
!      .. Executable Statements ..
Write (nout,*) 'F04JGF Example Program Results'
Write (nout,*)
!      Skip heading in data file
Read (nin,*)
Read (nin,*) m, n
tol = 5.0E-4_nag_wp
lda = m
lwork = 4*n
Allocate (a(lda,n),b(m),work(lwork))
Read (nin,*)(a(i,1:n),i=1,m)
Read (nin,*) b(1:m)

!      ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call f04jgfe(m,n,a,lda,b,tol,svd,sigma,irank,work,lwork,ifail)

Write (nout,*) 'Solution vector'
Write (nout,99997) b(1:n)
Write (nout,*)

```

```

      Write (nout,99999) 'Standard error = ', sigma, '      Rank = ', irank
      Write (nout,*)
      Write (nout,99998) 'SVD = ', svd

99999 Format (1X,A,F6.3,A,I2)
99998 Format (1X,A,L2)
99997 Format (1X,8F9.3)
      End Program f04jgfe

```

10.2 Program Data

```

F04JGF Example Program Data
  6  4                               : m, n
  0.05  0.05  0.25 -0.25
  0.25  0.25  0.05 -0.05
  0.35  0.35  1.75 -1.75
  1.75  1.75  0.35 -0.35
  0.30 -0.30  0.30  0.30
  0.40 -0.40  0.40  0.40           : matrix A
  1.0   2.0   3.0   4.0   5.0   6.0 : vector B

```

10.3 Program Results

F04JGF Example Program Results

```

Solution vector
  4.967  -2.833  4.567  3.233

Standard error =  0.909      Rank =  3

SVD =  T

```
